A Tutorial on Monochromatic Waves

Raul Estevez

In the present study we shall review the application of a technique of numerical extrapolation of monochromatic waves, to some theoretical models. The already known technique, described in Claerbout (1970a), (1970b) refers to the computation of vertical or quasi-vertical downgoing waves in a lightly inhomogeneous medium by means of finite difference approximations to the scalar acoustic wave equation. In our case, the technique was applied to study the downward continuation of a plane wave through a quasi-inhomogeneous medium with different two dimensional geometrical bodies of slightly different velocity embedded within it.

Although we are going to consider space variations of velocity, for simplicity let's assume that density is space independent and that we do not have conversions of pressure waves to shear waves (Actually, the more general case of variable density can be handled without any noticeable difficulty.). In this case, the equation for pressure waves can be written as

$$\nabla^2 P - \frac{1}{c^2} P_{tt} = 0 \tag{1-1}$$

where the velocity $c = \sqrt{k/\rho}$, k - incompressibility, ρ - density and $P_{tt} = \partial^2 P/\partial t^2$.

If c were constant, we know that equation (1) would admit plane wave solutions of the type

$$P = P_o \exp(ik_x x + ik_y y + ik_z z - i\omega t) = P_o \exp(i \vec{k} \vec{r} - i\omega t) . \tag{1-2}$$

By substituting (1-2) into (1-1) we easily find that P_o is a constant, provided that $c^2 = \omega^2/k^2$, where $k^2 = \Sigma k_i k_i = k_x^2 + k_y^2 + k_z^2$. Since in many situations of practical interest the waves do not propagate at great angles from the vertical, for simplicity we are going to further limit our discussion to a two-dimensional case (x,z-dependence only), where the waves are traveling in the plus z-direction. In this case $k_x = k_y = 0$ and $k_z = \omega/c = m$. The solution to (1-2) then becomes

$$P = P_o \exp(i m z - i \omega t) . \tag{1-3}$$

In a more general case, c may be an arbitrary function of space. By analogy with (1-3) let us try, for the downgoing wave, a solution of the type

$$P = Q^+ (x,z) \exp(i \bar{m} z - i \omega t) , \tag{1-4}$$

where, as we see, the function $Q^+(x,z)$ is no longer a constant, and the "+" sign indicates that the wave is traveling in the plus z-direction. The bar over $\bar{m}$ denotes that in this case we are considering a spatial average of the spatial frequency m .

$$m = \frac{\omega}{c(x,z)} \quad \text{and} \quad \bar{m} = \frac{\omega}{\bar{c}(x,z)}$$

Although function (1-4) may in general vary rapidly with z, if the inhomogeneities are small, at some distance from the source, the wave is going to behave as if it were approximately planar, which means that Q^+ will be a slowly variable function in z . This fact will be used later on, while it is worth mentioning that this assumption allows us to use a coarser grid to represent Q^+ , in relation to the finer grid that we may need to represent P .

Since we suppose that there may be velocity inhomogeneities, besides the downgoing wave, reflected waves will arise, and a more general solution should rather be written as

$$P = (Q^{+} \exp(i\bar{m}z) + Q^{-} \exp(-i\bar{m}z)) \exp(-i\omega t) = (Q^{+} + Q^{-} \exp(-i2\bar{m}z)) \exp(i\bar{m}z - i\omega t)$$

$$P = \tilde{Q}(x,z) \exp(i\bar{m}z - i\omega t)$$

Here, $\tilde{Q}$ becomes a rapidly varying function that would require a finer grid. If we could handle both waves separately, we would be able to describe them on coarser grids. This is one of the reasons why we do not want to treat both solutions on the same grid and, therefore, why we want to seek an equation describing only the transmitted waves and another to describe only the reflected waves. As we shall see later, discarding the second term in expression (1-5) can be justified on grounds of the slowly varying character of Q and the fact that in usual geophysical situations, transmitted waves are of much greater strength than reflected waves.

Computing Q^{+}_{zz} , Q^{+}_{tt} and replacing into the two dimensional wave equation (1-1), we get

$$Q^{+}_{zz} + Q^{+}_{xx} + 2 i \bar{m} Q^{+}_{z} + (m^{2} - \bar{m}^{2}) Q^{+} = 0 \qquad (1\text{-}6)$$

If, as we said, the amplitude modulation Q^{+} is a slowly varying function with z , then obviously $Q^{+}_{zz} << 2 im Q^{+}_{z}$ and we might drop the Q^{+}_{zz} term for computational convenience. Then we get a first order, and hence initial-value equation in z.

$$Q^{+}_{xx} + 2 i \bar{m} Q^{+}_{z} (m^{2} - \bar{m}^{2}) Q^{+} = 0 \qquad (1\text{-}7)$$

Notice that for a homogeneous medium $\bar{m} = m$, and therefore (1-7) reduces to

$$Q^+_{xx} + 2 i \bar{m} Q^+_z = 0 \tag{1-8}$$

The considered approximation $Q^+_{zz} \approx 0$, as we shall see later, in particular suppresses the reflected waves in the equation.

If we define now

$$E(x,z) = \frac{\omega^2}{\bar{m}^2 c^2(x,z)} - 1 = \frac{m^2}{\bar{m}^2} - 1 , \tag{1-9}$$

equation (1-7) can be rewritten as

$$Q^+_{xx} + 2 i \bar{m} Q^+_z + E(x,z) \bar{m}^2 Q^+ = 0 \tag{1-10}$$

This new parameter E is quite important, since through it we are going to introduce the inhomogeneities of the medium: in the case of a homogeneous medium ($\bar{m} = m$) $E = 0$, but in regions of space (grid) where some inhomogeneity is present, then we choose $\bar{m}$ in such a way that the spatial average of E is kept small, since our approximations only hold for small departures from a homogeneous situation.

Let us finish the section by adding some additional remarks on the character of the approximation $Q^+_{zz} \simeq 0$, since this assumption will be responsible for the major limitations of this technique.

The first thing we would like to notice is the connection between this term and the reflected wave solution. Once Q^+ has been calculated, in order to compute Q^- we insert the complete trial solution (1-5) into the wave equation (1-1) and subtract equation (1-7), corresponding to the transmitted waves. After dropping Q^-_{zz} again, we get

$$Q^-_{xx} - 2 i \bar{m} Q^-_z + E(x,z) \bar{m}^2 Q^- = -Q^+_{zz} \exp(i 2 \bar{m} z) . \tag{1-11}$$

We notice that this equation is almost like equation (1-7) for transmitted waves, but differs in the sign of the term containing i (which changes if we consider the change of the sign of z), and in the presence of a source-like term, proportional to Q^{+}_{zz} . From here we see that by neglecting the Q^{+}_{zz} term in equation (1-6), in practice we just suppressed the reflections. It is not difficult to realize that, on the other hand, the presence of the Q^{+}_{zz} and Q^{-}_{zz} terms, would give rise to higher order reflections (multiples).

The second aspect that we would like to refer in relation with the approximation is its meaning in the domain of the anti-transformed variable P , since it would allow us to compare directly the approximated equation (1-7) with the actual wave equation (1-1) . For simplicity, let us consider the case of a homogeneous medium; then, if according to (1-4) we now make

$$Q^{+}(x,z) = P(x,z) \exp(-i\bar{m}z) , \tag{1-12}$$

compute Q^{+}_{z} , Q^{+}_{xx} and replace into (1-8), we get

$$P_{xx} + 2\, i\, \bar{m}\, P_{z} + 2\, \bar{m}^{2}\, P \;=\; P_{xx} + 2\, i\, m\, P_{z} + 2\, m^{2}\, P \;=\; 0 \tag{1-13}$$

The wave equation (1-1) in two dimensional cartesian coordinates is

$$P_{xx} + P_{zz} - m^{2}\, P \;=\; 0 \tag{1-14}$$

Let us now try in both equations a plane wave-type solution of unit magnitude and propagating in the $k = (k_{x}, k_{z})$ - direction

$$P \;=\; \exp(i\, k_{x}\, x + i\, k_{z}\, z) \tag{1-15}$$

Inserting this trial solution into both (1-13) and (1-14) leads to two algebraic equations for k_x and k_z, usually known as dispersion relations. They are respectively

$$k_x^2 + 2\,m\,k_z - 2\,m^2 = 0 \qquad (1\text{-}16)$$

and

$$k_x^2 + k_z^2 - m^2 = 0 \qquad (1\text{-}17)$$

The graphs, corresponding to these two relations, follows (Fig. 1).



Fig. 1. Graph of the dispersion relations for the wave equation (circle in solid line) and for the one-way wave equation (parabola in dotted line).

As we see, the dispersion relation for the wave equation corresponds to the well known circle of radius $m = \omega/c$ or $(k_x^2 + k_z^2)^{1/2} = m$, while the relation for the one-way wave approximation corresponds to a parabola, that fits the circle only for small values of the angle θ. This graph also shows how, in the case of the wave equation, that for any value of k_x, we get two values of k_z, corresponding to waves travelling in two opposite directions, while in the considered approximation, we get only one value for k_z. For this reason, the approximation $Q^+_{zz} = 0$ is often called the parabolic approximation, and, from this point of view, it limits the validity of our results to waves travelling within small angles of the z-axis.

Actually, Claerbout (1970b) showed that this approximation is just a special case of a more general and precise one.

$$k_z = (m^2 - k_x^2)^{1/2} \approx m \frac{4\ m^2 - (n+1) k_x^2}{4\ m^2 - (n-1) k_x^2} , \quad n = 1,2,3,\ldots , \tag{1-18}$$

In the P domain, approximation (1-18) is equivalent to

$$(m^2 + \partial_{xx})^{1/2} \approx m \frac{4\ m^2 + (n+1) \partial_{xx}}{4\ m^2 + (n-1) \partial_{xx}} , \tag{1-19}$$

which for the case of $n = 1$, originates the approximated equation

$$Q_z = \frac{i}{2\ k} [Q_{xx} + \frac{2}{1+k_z/m} k_z^2 \ E\ Q] \tag{1-20}$$

For a homogeneous medium, $E = 0$ and $k_z \equiv \bar{m} = m$, thus (1-20) becomes

$$Q_z = \frac{i}{2\ m} Q_{xx} , \tag{1-21}$$

which coincides with (1-8). For the more general case of a inhomogeneous medium, (1-20) does not coincide with (1-10), but we may notice that if E is small enough so that $\bar{m} = k_z \sim m$, equation (1-20) becomes

$$Q_z = \frac{i}{2\ \bar{m}} [Q_{xx} + \bar{m}^{-2} \ E\ Q] , \tag{1-22}$$

which exactly coincides with (1-10).

A more precise approximation corresponds to the case $n = 2$.

$$Q_z = \frac{i}{2\,m}\left(1 - \frac{\bar{m}}{2\,m}\right)\left[\,Q_{xx} + \frac{4m^2\,\bar{m}^2}{(m+\bar{m})(2m-\bar{m})}\,E\,Q\,\right] - \frac{1}{4m^2}\,Q_{xxz}\;. \tag{1-23}$$

Nevertheless, we are going to restrict our further considerations to a simpler and less precise approximation, close to (1-23). Other kinds of approximations, and specifically, a hyperbolic type approximation will be briefly discussed in the section corresponding to slant frames.

<u>The Computer Algorithms. The explicit, implicit and mixed schemes.</u>

In order to apply the finite difference technique, the usual procedure is to solve equation (1-7) for Q_z^+ .

$$Q_z^+ = \frac{i}{2\,\bar{m}}\left(\,Q_{xx}^+ + E\,\bar{m}^2\,Q^+\,\right)\;. \tag{2-1}$$

If now we want a better approximation to the actual wave equation, we could compute Q_{zz}^+ from (2-1) and replace into (1-6). In the P-domain, this would correspond to a better parabolic fit to the circle of Fig. 1, allowing us to consider a wider range of angles θ . In terms of the general approximation (1-18), what we intend to do comes out to be close to the case $n = 2$.

Calculating then Q_{zz}^+ from (2-1), dropping higher derivatives of Q^+ in relation to z (Q_{zzz}^+) and inserting in (1-6), we get

$$\left(1 + \frac{E}{4}\right) Q_z^+ = \frac{i}{2\,\bar{m}}\left[\,Q_{xx}^+ + \left(E + \frac{i\bar{m}}{2}\,E_z\right)\bar{m}^2\,Q^+\,\right] - \frac{1}{4\bar{m}^2}\,Q_{xxz} \tag{2-2}$$

In the concrete examples, where we will apply this equation, we are going to consider E to be constant all throughout the regions of inhomogeneities, thus we are going to discard the term containing E_z .

$$(1 + \frac{E}{4}) Q_z^+ = \frac{i}{2\bar{m}} [Q_{xx}^+ + E \bar{m}^2 Q^+] - \frac{1}{4\bar{m}^2} Q_{xxz} \qquad (2\text{-}3)$$

By applying the finite difference method to equations (2-1), (2-2) or (2-3), we just try to get a recursive relationship such that, given Q^+ at some fixed value of z for all x , we could extrapolate downward the solution to any other value of z .

Since we are going to represent Q^+ in a grid, let's start by defining the following difference notation

$$Q^+(x,z) = Q_{m(\Delta x)}^{n(\Delta z)} \quad (n,m - \text{integers}) , \qquad (2\text{-}4)$$

where n will account for the variation in z , and m for the variation in x (Starting from now we will omit the "+" superscript). According to this notation, we then approximate the first derivative of Q in the (n,m)-point of the grid, through the difference

$$Q_z \cong \frac{Q(x,z+\Delta z)-Q(x,z)}{\Delta z} = \frac{Q_m^{n+1} - Q_m^n}{\Delta z} \qquad (2\text{-}5)$$

The approximation to the second derivative of Q in x (Q_{xx}) , requires a little more consideration. In effect, since the considered equations deal with the first and second derivatives of Q : Q_z and Q_{xx} , in order to evaluate them in the point (n,m) of the grid, we have to take into consideration as well, the neighbor points: (n,m-1) ; (n,m+1) ; (n+1,m-1); (n+1, m) and (n+1,m+1) (see Fig 2).



Fig. 2. Representation of the Grid. We want to evaluate Q_z and Q_{xx} in the point (n,m) .

Thus, it seems that we have two different choices in order to define Q_{xx} . The most evident and used one is

$$Q_{xx}^{(1)} \cong \frac{Q_{m+1}^{n} - 2Q_{m}^{n} + Q_{m-1}^{n}}{\Delta x^2} , \qquad (2\text{-}6)$$

which makes use of the neighbor points of (n,m) , along the nth row, and the less common approximation,

$$Q_{xx}^{(2)} \cong \frac{Q_{m+1}^{n+1} - 2\, Q_{m}^{n+1} + Q_{m-1}^{n+1}}{\Delta x^2} , \qquad (2\text{-}7)$$

which makes use of the corresponding points on the next row. The first approximation gives rise to the so-called "explicit scheme", while the second one gives rise to the so-called "implicit scheme". The idea of also taking into consideration (2-7), besides (2-6), is connected with the fact that the first derivative of Q in z , is centered at (n+1/2,m) , between rows n and n + 1 . With all these three approximated derivatives (2-5),(2-6) and (2-7) centered at different locations on the grid, we might expect some troubles, especially when considering short wavelengths. In fact, if we introduce the trial solution $Q_m^n = Q^n e^{ikm}$ of a plane wave propagating at small angle from the z-axis into equation (2-1), it can easily be shown that, in the case of the explicit scheme, the magnitude of Q incorrectly increases by a factor of $\left|1 + i\,\frac{\bar{m}\,E\,\Delta z}{2}\right|$ for each step in the z-direction, while off-axis waves are amplified unless Δz is arbitrarily small. In the case of the implicit scheme, we get opposite effects. Q decreases by the same factor in each step, while off-axis waves are attenuated.

One could think of a solution to this problem being, centering the first derivative Q_z at (n,m), according to

$$Q_z \approx \frac{Q_m^{n+1} - Q_m^{n-1}}{2\,\Delta z} \; ; \qquad \text{(2-8}$$

but it turns out to be even worse. It creates instability for any Δx, due to the fact that, by defining this derivative over two steps in z, we artificially convert our first order differential equation in z into a second order equation.

The other possible solution is to center the second derivative Q_{xx} also at $(n+1/2,m)$. This can be achieved by taking the average between (2-6) and (2-7), or by computing the second derivative of $(Q_m^n + Q_m^{n+1})/2$ (both ideas lead to the same result):

$$Q_{xx} \approx \frac{Q_{xx}^{(1)} + Q_{xx}^{(2)}}{2} \qquad \text{(2-9)}$$

It turns out that this last scheme, called the "mixed" or Crank-Nicolson scheme, guarantees stability for any Δx and can also be applied to similar situations in electromagnetics and elasticity. Introducing a parameter θ, we can summarize all these three schemes in the relation

$$Q_{zz} \approx \frac{1}{2(\Delta x)^2} (Q^{n+1}_{m+1} - 2Q^{n+1}_{m} + Q^{n+1}_{m-1})\theta + \frac{1}{2(\Delta x)^2} (Q^{n}_{m+1} - 2Q^{n}_{m} + Q^{n}_{m-1})(1-\theta) \quad , \tag{2-10}$$

where if $\theta = 1$, we have the implicit scheme; if $\theta = 0$, we have the explicit scheme; and if $\theta = .5$ we have the mixed scheme. Varying θ , relation (2-10) can be used to filter the solution in case that for any reason we wanted to amplify or to attenuate off-axis waves.

In order to simplify notation, we will introduce the second derivative operator in x , according to

$$T_{xx}(Q^n_m) = Q^n_{m+1} - 2 Q^n_m + Q^n_{m-1} \ . \tag{2-11}$$

In terms of this notation, the approximation (2-9) could be written as

$$Q_{xx} \approx \frac{1}{2(\Delta x)^2} T_{xx}(Q^{n+1}_m + Q^n_m) \tag{2-12}$$

Better approximations to the second derivative, in terms of the T_{xx} operator, can be found. For example, if we make use of the Taylor's Series expansion for $Q(x+\Delta x,z)$ and $Q(x-\Delta x,z)$:

$$Q(x+\Delta x,z) = Q(x,z) + \Delta x\, Q_x(x,z) + \frac{\Delta x^2}{2} Q_{xx}(x,z) + \cdots \tag{2-13}$$

and

$$Q(x-\Delta x,z) = Q(x,z) - \Delta x\, Q_x(x,z) + \frac{\Delta x^2}{2} Q_{xx}(x,z) + \cdots \tag{2-14}$$

then, according to (2-11), we can write

$$T_{xx}Q = Q(x+\Delta x,z) - 2Q(x,z) + Q(x-\Delta x,z) \cong \Delta x^2\, Q_{xx}(x,z) + \frac{\Delta x^4}{12} Q_{xxxx}(x,z) + \ldots \tag{2-15}$$

But since $T_{xx} Q \cong \Delta x^2 Q_{xx}$, we can rewrite (2-15) as

$$T_{xx} Q \cong \Delta x^2 (1 + \frac{1}{12} T_{xx}) Q_{xx}$$

or

$$Q_{xx} \approx \frac{1}{\Delta x^2} \frac{T_{xx}}{1+(1/12)T_{xx}} Q \tag{2-16}$$

In terms of the mixed scheme, expression (2-16) would have to be written as

$$Q_{xx} \approx \frac{1}{2 \Delta x^2} \frac{T_{xx}}{1+(1/12)T_{xx}} (Q_m^{n+1} + Q_m^n) \tag{2-17}$$

The operator $1 +(1/12)T_{xx}$ in the denominator has to be understood as a factor that multiplies all the other elements of the equation, where Q_{xx} appears. Approximation (2-17) improves the results and, in particular, allows us to sample the data in bigger steps than stability usually requires (approximately by a factor of 2) . Nevertheless, for simplicity, we are going to use approximation (2-12) in the following computations.

Applying the mixed scheme to equation (2-3), we then get

$$(1+ \frac{E}{4}) \frac{Q_m^{n+1} - Q_m^n}{\Delta z} = \frac{i}{4\bar{m}(\Delta x)^2} [T_{xx} (Q_m^{n+1} + Q_m^n) + E \bar{m}^2 (\Delta x)^2 (Q_m^{n+1} + Q_m^n)] -$$

$$- \frac{1}{4\bar{m}^2 (\Delta x)^2 \Delta z} T_{xx} (Q_m^{n+1} - Q_m^n) \tag{2-18}$$

In order to see how we can get the $n + 1$ z-level from the n-level, let's bring all the $n + 1$ terms to the left and the n terms to the right.

$$a\left[\frac{(4+E)b^2}{1+c^2}(1+ic)+T_{xx}-\frac{Eb^2c}{1+c^2}(c-i)\right]Q_m^{n+1}$$

$$=a^*\left[\frac{(4+E)b^2}{1+c^2}(1-ic)+T_{xx}-\frac{Eb^2c}{1+c^2}(c+i)\right]Q_m^n\ , \tag{2-19}$$

where

$$b=\bar{m}\,\Delta x\ ;\ c=\bar{m}\,\Delta z\ ;\ a=\frac{1}{b\,\Delta x}\left(\frac{1}{c}-i\right)\quad\text{and}\quad a^*=\text{conj}(a) \tag{2-20}$$

According to (2-11), this expression gives rise to the tri-diagonal set of equations

$$\begin{bmatrix} A & 1 & 0 & 0 & \cdots & 0 \\ 1 & A & 1 & 0 & \cdots & 0 \\ 0 & 1 & A & 1 & \cdots & 0 \\ \cdot & \cdot & \cdot & \cdot & \cdots & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdots & \cdot \\ 0 & \cdot & \cdot & 1 & A & 1 \\ 0 & \cdot & \cdot & 0 & 1 & A \end{bmatrix} \begin{bmatrix} Q_1 \\ Q_2 \\ \cdot \\ \cdot \\ \cdot \\ \cdot \\ Q_m \end{bmatrix}^{n+1} = \frac{a^*}{a} \begin{bmatrix} A^* & 1 & 0 & 0 & \cdots & 0 \\ 1 & A^* & 1 & 0 & \cdots & 0 \\ 0 & 1 & A^* & 1 & \cdots & 0 \\ \cdot & \cdot & \cdot & \cdot & \cdots & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdots & \cdot \\ 0 & \cdot & \cdot & 1 & A^* & 1 \\ 0 & \cdot & \cdot & 0 & 1 & A^* \end{bmatrix} \begin{bmatrix} Q_1 \\ Q_2 \\ \\ \\ \\ \\ Q_m \end{bmatrix}^{n}\ , \tag{2-21}$$

where

$$A=\frac{4(\bar{m}\,\Delta x)^2}{1+(\bar{m}\,\Delta z)^2}+E(\bar{m}\,\Delta x)^2-2+i\,\frac{4(\bar{m}\,\Delta x)^2(\bar{m}\,\Delta z)}{1+(m\,\Delta z)^2}\ ;\ A^*=\text{conj}(A), \tag{2-22}$$

and

$$\frac{a^*}{a}=\frac{1-(\bar{m}\,\Delta z)^2}{1+(\bar{m}\,\Delta z)^2}+i\,\frac{2(\bar{m}\,\Delta z)}{1+(\bar{m}\,\Delta z)^2}\ . \tag{2-23}$$

It is worth noticing that the first and the last rows of the matrices in (2-16) have to be filled out separately and according to the chosen bordering conditions. In our case we have chosen zero slope condition. In a more general case, the upper and lower diagonals may not be equal.

A set of equations like (2-16) can be solved extremely easily, and a fast method of solution is described in Claerbout (1970a) and (1973). The corresponding computer program (subroutine C TRI) is included as a part of a more general example program in one of the appendices. The usual procedure to follow is, that given an initial string of values (source) corresponding to the first row ($n = 1$ or $z = 0$), we use (2-16) to extrapolate the solution up to any n (or z).

The grid:

Let us spend a few words in the grid itself and some of the criteria we must look at when choosing the intervals of sampling. When we approximate the derivatives in x, z or t through finite differences, it happens that the short wavelengths tend to attenuate or amplify depending on the sign of the numerical viscosity. One of the reasons for this behavior is the fact that by doing these approximations, in practice, we make use of the long wavelength approximations $\omega \Delta t \cong 2 \tan(\omega \Delta t/2)$ and $k_x \Delta x \cong 2 \sin(k_x \Delta x/2)$. For example, when we sample in time, we first replace the time derivative by $-i\omega$, and then re-express ω in terms of the z-transform variable $z = \exp(i \omega \Delta t)$:

$$i \omega \Delta t = \ln z = 2\left[\frac{z-1}{z+1} + \frac{1}{3} \left(\frac{z-1}{z+1} \right)^3 + \cdots\right] \approx 2 \frac{z-1}{z+1} = i\, 2 \tan \frac{\omega \Delta t}{2} \qquad (3\text{-}1)$$

The kind of approximation we make by retaining only the first term of this expansion is shown in fig. 3.



Fig. 3. z-transform approximation to the time derivative.

More or less a similar situation is found when approximating the derivatives in x or z through finite differences (in this case, as we said before, the approximation is of the type $k_x \Delta x \approxeq 2 \sin(k_x \Delta x/2)$).

Since we can connect $\omega \Delta t$, $k_x \Delta x$ and $k_z \Delta z$ with the number of points per wavelength in time, N_t , or in either direction, N_x and N_z , through the relations

$$N_t = \frac{\lambda}{\Delta t} = \frac{2\pi c}{\omega \Delta t} \quad ; \quad N_x = \frac{\lambda x}{\Delta x} = \frac{2\pi}{k_x \Delta x} \quad ; \quad N_z = \frac{\lambda z}{\Delta z} = \frac{2\pi}{k_x \Delta z} , \tag{3-2}$$

we then restrict the validity of our results to waves sampled moderately densely. The relative error in terms of points per wavelengths is tabulated in fig. 4. On the basis of this table and choosing an acceptable error, say 3% - 5% , one determines a minimum acceptable number of points per wavelength which comes close to 8-10 points per wavelength. Now, since $k_x^2 + k_z^2 = \omega^2/c^2$, for freely propagating waves we have that $k_x , k_z < \omega/c$. Thus, the Fourier transform $Q(k_x \omega)$ should vanish unless $k_x < \omega/c$. With the assumption of 10 points per wavelength, the useful bandwidth $- 2\pi/10 < \omega \Delta t < + 2\pi/10$ is then markedly less than the total bandwidth available (the 2π periodicity interval for transforms of sampled data), being the ratio of useful bandwidth to total bandwidth equals 1/5 . Fig. 5 shows the usable portion of (ω, k_x) space.

In the computations of the present study, a minimum value of 8 points per wavelength was chosen. Using then relation (3-2), this allows us to choose a value for Δz :

$$\Delta z = \frac{2\pi}{\bar{m} \cdot N_z} = \frac{\pi}{4\bar{m}} \tag{3-3}$$

$\omega\Delta t$ or $k_x\ \Delta X$, radians	Points per wavelength, $2\pi/\omega\Delta t$	Relative error of $2\tan\omega\Delta t/2$	Relative error of $2\sin k_x\Delta x/2$	
2×10^{-n}	$\pi\times10^{n}$	$10^{-2n}/3$	$10^{-2n}/6$	$n\geqslant1$
.31416	20	.8%	.4%	
.3926	16	1.2%	.6%	
.5235	12	2.3%	1.1%	
.6283	10	3.4%	1.6%	
.7853	8	5.4%	2.5%	
1.047	6	10.0%	4.4%	
1.571	4	27.0%	9.0%	

Figure 4. The relative error at short wavelengths often associated with expressing differential equations in difference form.



Figure 5. Useful (good) and non-useful portions of frequency-wave number space. This figure depicts the usual case in observations; the inequality $\Delta x > c\Delta t$ usually arises because extra points in time are often more cheaply obtained than extra points in space.

Another interesting parameter, connected with the dimensions of the grid, is the total number of wavelengths that we will have in a given grid. If we call NZ , the total number of points in the z-direction, LZ - the total corresponding length and N_λ - the total number of wavelength in LZ , then we have

$$N_\lambda = \frac{LZ}{\lambda} = \frac{NZ + \Delta Z}{N_z * \Delta Z} = \frac{NZ}{N_z} , \qquad (3\text{-}4)$$

and we get the curious result that N_λ does not depend directly on the chosen spacing, but rather on the total number of points NZ and the chosen number of points per wavelength N_z .

Although relation (3-3) gives us a criterion in order to choose the spacing, it is worth noticing again that this criterion refers to the full solution P and not to the amplitude modulation Q . So that, if the inhomogeneities are not big (E - is kept small enough), and therefore Q is a slowly variable function, we may compute Q in a coarser grid and then, through some kind of interpolation, we can make the much more easy computation of P (we just have to multiply Q by a phase term) on a finer grid, defined by relation (3-3).

In order to illustrate the whole procedure, a detailed program was written and it is attached at the end. The program was used to compute the propagation of plane and circular waves through different two-dimensional bodies embedded within a homogeneous medium. At the end, a short illustrative movie was made for the case of a prismatic body. In this last case, several different frequencies were summed in an intent to pick up some time-dependent effects.



Figure 6. Plane waves of four different frequencies, propagating through a right 45° prism. Due to the superposition of equally spaced frequencies, diffractions from the right upper corner can easily be seen (A). The movie was based on this figure.



Figure 7. Plane waves propagating through a slow velocity prism (inverted version of fig. 6). Notice the circular diffractions from the upper corner of the prism (A) and, as in the previous fig., the shadow zone beside its vertical side (B).



Figure 8. Circular waves impinging on the same prism of fig. 7. Toward the sides of the page, reflections from the primary circular wave are noticeable (A), while in the center, inside the prism, both waves mix with a smaller wavelength, due to the lower velocity of the prism (B).



Figure 9. Plane waves impinging on a plano-convex lense of lower velocity. The lense changes the initially plane waves (A) into circular convergent waves (B).



Figure 10. Circular waves illuminating the same plano-convex lense of fig. 9. Here the lense straightens up the circular waves. As in the case of fig. 8, side reflections mix and interfere with the initial wave.

```
C        THIS PROGRAM EXTRAPOLATES DOWNWARD A TWO DIMENSIONAL PLANE WAVE
C        P(X,Z), THROUGH A MEDIUM WITH DIFFERENT GEOMETRICAL BODIES EMBEDED
C        WITHIN IT. IT ALLOWS ALSO TO SUPERIMPOSE SEVERAL DIFFERENT FRE-
C        QUENCIES (K). THE OUTPUT OF THE COMPUTATION (WAVE FIELD P(X,Z))
C        CAN BE PLOTTED WITH THE PRINTER PLOTTER (SUB. OUT), WITH THE VERSATEK
C        PLOTTER (SUB. VOUT) OR STORED ON A TAPE (SUB. TOUT). THE OUTPUT
C        IS OF THE SIZE OF A 70 X 100 GRID BUT, IN THE COMPUTATION, LARGER
C        GRIDS MAY BE USED AS WELL, MULTIPLYING THESE TWO NUMBERS BY ANY
C        INTEGER. IN THE PRESENT EXAMPLE A GRID OF 280 X 400 POINTS IS USED
C        AND ACTUALLY THE PROGRAM INCORPORATES THE USE OF A TEMPORARY DISK
C        (UNIT 40) THAT PERMITS TO HANDLE LARGE GRIDS AND MANY DIFFERRENT
C        FREQUENCIES. "AMP" IS A TEMPORARY CONSTANT WHICH IS USED TO REDUCE
C        THE OUTPUT TO THE SPECIFIED SIZE. IN THE OUTPUT SUBROUTINES, THE
C        DATA IS SCALED BY EQUATING THE BIGGEST VALUE MULTIPLIED BY THE PARAMET
C        "CLIP", TO THE CLIPPING VALUE OF THE PLOTTERS(1024.)."NP(K)" IS THE
C        NUMBER OF SAMPLE POINTS PER WAVELENGTH CORRESPONDING TO FREQUENCY
C        K. "FREQ" IS THE NUMBER OF DIFFERENT FREQUENCIES TO BE SUPERIMPOSED
C        "DX" , "DZ" ARE THE SAMPLE INTERVALS IN THE X AND Z DIRECTIONS
C        (AFTER THE INTERPOLATION), WHILE "DZINT" IS THE INTERVAL BETWEEN
C        SUCCESSIVE COMPUTED VALUES OF Q (BEFORE THE INTERPOLATION). THE
C        LAST THREE VARIABLES ARE GENERATED BY THE PROGRAM AS A FUNCTION OF
C        NP, M, VEP AND INT. "M" IS THE WAVE NUMBER IN THE Z-DIRECTION.
C        "E" - THE FUNCTION OF THE INHOMOGENEITIES. "F(K)" ARE THE CHOSEN
C        DIFFERENT FREQUENCIES(IN CPS) TO BE SUPERIMPOSED. ONLY F(1) IS FIXED A
C        THE PROGRAM GENERATES "FREQ" EQUIDISTANT FREQUENCIES BETWEEN F(1)
C        AND 0. "X" , "Y" , "XS" , "XSS" ARE CONSTANTS TO BE USED IN SUB.
C        MEDIUM IN CONNECTION WITH THE COMPUTATION OF THE MEAN DIAGONAL
C        "CDIAG" OF SUB. CTRI. "ZA" , "ZB" , "CS" , "CSUB" , "CGAMMA" AND
C        "CBETA" ARE CONSTANTS TO BE USED IN SUB. CTRI. "CQINP" AND "CQOUT"
C        ARE THE INPUT AND OUTPUT ROWS IN SUB. CTRI. "SLOPE" KEEPS THE SLOPES
C        BETWEEN TWO SUCCESSIVE COMPUTED VALUES OF Q (CQINP AND CQOUT) IN
C        ORDER TO LINEARLY INTERPOLATE "INT" (TO BE FIXED) VALUES BETWEEN
C        THEM. "CE" , "CW" AND "CWA" ARE CONSTANTS TO BE USED IN ORDER TO
C        TRANSFORM BACK Q INTO P. "VA" IS THE VELOCITY OF THE MEDIUM (TO
C        BE FIXED) AND "VB" IS THE VELOCITY INSIDE THE INHOMOGENEITY (CONS-
C        TANT IN OUR CASE AND GENERATED BY THE PROGRAM AS A FUNCTION OF "E").
C        "WL" IS THE WAVELENGTH. "VEP" IS THE VERTICAL EXAGERATION OF THE
C        PRINTER PLOT. "NX" AND "NZ" ARE THE NUMBER OF POINTS IN THE X AND
C        Z DIRECTIONS RESPECTIVELY. THE INTEGER "K" WILL ALWAYS BE ASSOCIATED
C        WITH THE NUMBER OF FREQUENCIES THAT ARE SUPERIMPOSED. "I" WILL BE
C        ASSOCIATED WITH THE NUMBER OF POINTS IN THE Z-DIRECTION AND "J"
C        WITH THE NUMBER OF POINTS IN THE X-DIRECTION. "NF" IS THE NUMBER
C        OF FRAMES, CORRESPONDING TO SUCCESSIVE INSTANTS OF TIME, STARTING
C        FROM T=0."PARR(NX)" IS A TEMPORARY ARRAY USED TO WRITE AND READ ON
C        THE DISK."DT" IS THE PERIOD OF THE HIGHEST FREQUENCY (F(1)) AND
C        "TIMINT"- THE FRACTION OF THIS PERIOD THAT WE WANT BETWEEN TWO SUC-
C        CESSIVE FRAMES. IN CHANGING THE GRID SIZE OR THE NUMBER OF FRE-
C        QUENCIES, NOTICE THAT THE DIMENSIONS OF THE USED ARRAYS ARE: NP(K),
C        M(K),F(K),X(K),Y(K),XS(K),XSS(K),ZA(K),ZB(K),SLOPE(NXOUT,K),
C        CQINP(NX,K),CQOUT(NX,K),P(NXOUT,NZ),CDIAG(NX),CSUB(NX),CE(K),
C        CGAMMA(NX),CBETA(NX),CS(K),CW(K),CWA(K); WHERE K IS THE NUMBER
C        OF FREQUENCIES TO BE SUPERIMPOSED AND NXOUT IS THE NUMBER OF
C        COLUMNS(TRACES) IN THE OUTPUT (70 IN OUR CASE).
C
```

```
C
C
      IMPLICIT COMPLEX (C)
      INTEGER NP(1),FREQ,AMP
      INTEGER*2 IHW(400)
      REAL DX,DZ,M(1),E,P(70,400),F(1),X(1),Y(1),XS(1)
      REAL XSS(1),ZA(1),ZB(1),SLOPE(70,1),CLIP
      COMPLEX PARR(70)
      COMPLEX CQINP(280,1),CQOUT(280,1),CDIAG(280),CSUB(280),CE(1)
      COMPLEX CMPLX,CBETA(280),CGAMMA(280),CS(1),CW(1),CWA(1)
      NAMELIST /PLOT/NX,NZ,DX,DZ,VA,VB,F,M,DM,DT,E,DZINT
      NX=280
      NZ=400
      NXOUT=70
      CI=(0.,1.)
      VEP=10./6.
      AMP=NX/70
      REWIND 40
C
C     PARAMETERS TO BE DEFINED
C
      CLIP=2.
      FREQ=1
      NP(1)=8
      NF=1
      INT=2
      E=.3
      F(1)=40.
      TIMINT=.2
      VA=3000.
C
C     THE FOLLOWING STATEMENTS ARE CONNECTED WITH THE COMPUTATION
C     OF DIFFERENT CONSTANTS TO BE USED IN THE MAIN PROGRAM AND IN
C     SOME OF THE SUBROUTINES.
C
      VB=SQRT(VA**2/(1.+2.*E))
      W=6.28*F(1)
      DT=(1./F(1))*TIMINT
      M(1)=W/VA
C
      DM=M(1)/FREQ
C
      WL=6.28/M(1)
      DW=VA*DM
      DZ=6.28/(NP(1)*M(1))
      DZINT=DZ*INT
C
C     WHEN USING PRINTER PLOT (SUB. OUT) MAKE DX=DZ/VEP
C
      DX=DZ
C
C     PARAMETERS CONNECTED WITH THE SUB. VOUT (VERSATEK)
C
      VERTEX=DX/DZ
```

```
      VXOUT=1./AMP
C
C
      DO 8 K=1,FREQ
      K1=K-1
      M(K)=M(1)-K1*DM
C
C
      CW(K)=CEXP(-CI*W*DT)
      NP(K)=6.28/(DZ*M(K))
      CE(K)=CEXP(CI*M(K)*DZ)
      F(K)=W/6.28
      X(K)=(4.*(M(K)*DX)**2)/(1.+(M(K)*DZINT)**2)
      Y(K)=X(K)*M(K)*DZINT
      X(K)=X(K)-2.
      XS(K)=X(K)+E*(M(K)*DX)**2
      XSS(K)=X(K)-E*(M(K)*DX)**2
      ZA(K)=(1.-(M(K)*DZINT)**2)/(1.+(M(K)*DZINT)**2)
      ZB(K)=(2.*M(K)*DZINT)/(1.+(M(K)*DZINT)**2)
      CS(K)=CMPLX(ZA(K),ZB(K))
      PRINT10,M(K),NP(K),F(K)
   10 FORMAT(' ','M=',F6.4,4X,'POINTS PER WAVELENGTH:',I4,4X,'FREQUENCY'
     1,F6.2,/)
    8 W=W-DW
C
C
C
C
C
C     NOW WE GENERATE THE FIRST STRING OF VALUES OF Q (WAVEFORM:
C     LINEAR OR CIRCULAR)
C
      CALL INPUT 1(NX,CQINP(1,1))
C
      DO 5 K=1,FREQ
      DO 5 J=1,NX
    5 CQINP(J,K)=CQINP(J,1)
C
C
C
C
C     STARTS THE BIG LOOP TO COMPUTE SUCCESSIVE ROWS OF THE GRID
C
      NZ1=NZ+1
      DO 120 I=1,NZ1,INT
C
C     COMPUTE THE ZTH ROW, CORRESPONDING TO THE KTH FREQUENCY
C
      DO  80 K=1,FREQ
C
C
C     DEFINE THE GEOMETRY OF THE INHOMOGENEITY.
C
      CALL MEDIUM 2(E,M(K),NX,DX,DZINT,I,CDIAG,X(K),Y(K),XS(K),XSS(K),
```

```
     1AMP)
C
C     SOLVE THE TRIDIAGONAL SET OF EQUATIONS
C
      CALL CTRI(NX,CDIAG,CQINP(1,K),CQOUT(1,K),CSUB,CS(K),CBETA,CGAMMA)
C
C     COMPUTE THE SLOPE BETWEEN TWO SUCCESSIVE ROWS OF Q FOR DIF-
C     FERENT FREQUENCIES K IN ORDER TO MAKE A LINEAR INTERPOLATION.
C
      L=0
      DO 80 J=AMP,NX,AMP
C
      L=L+1
   80 SLOPE(L,K)=(CQOUT(J,K)-CQINP(J,K))/DZINT
C
C     INTERPOLATE INT-VALUES BETWEEN CQINP AND CQOUT AND WRITE
C     THE COMPUTED AND THE INTERPOLATED ROWS ON A TEMPORARY DISK.
C
      DO 100 K1=1,INT
      DO 100 K=1,FREQ
      CONS=CE(K)**(I+K1-2)
      L=0
      DO 95 J=AMP,NX,AMP
      L=L+1
C
C
   95 PARR(L)=(CQINP(J,K)+SLOPE(L,K)*DZ*(K1-1))*CONS
  100 WRITE(40) PARR
      DO 120 K=1,FREQ
      DO 120 J=1,NX
  120 CQINP(J,K)=CQOUT(J,K)
C
C
C
C
C     TRANSFORM BACK TO P AND COMPUTE NF DIFFERENT FRAMES, CORRES-
C     PONDING TO SUCCESSIVE TIME INSTANTS.
C
      DO 250 N1=1,NF
      DO 150 I=1,NZ
      DO 150 J=1,70
  150 P(J,I)=0.
C
C
      REWIND 40
      N=N1-1
      DO 300 I=1,NZ
      DO 300 K=1,FREQ
      CWA(K)=CW(K)**N
      READ(40) PARR
      DO 300 J=1,70
  300 P(J,I)=PARR(J)*CWA(K)+P(J,I)
C
C
```

```
      PRINT30,VA,VB,E,WL,NX,NZ,DX,DZ,INT
   30 FORMAT(' ','OUTER VEL:',F8.2,'MT/SEG',4X,'INNER VEL:',F8.2,'MT/SEG
     1',4X,'E=',F5.2,4X,'WAVELENGTH:',F8.4,'MT',/,' NX=',I4,4X,'NZ=',I4,
     24X,'DX=',F6.4,'MT',4X,'DZ=',F6.4,'MT',4X,'POINTS INT.:',I4,//)
      WRITE(6,PLOT)
C
C     CALL THE WANTED OUTPUT : OUT, VOUT OR TOUT
C
  250 CALL TOUT(CLIP,NXOUT,NZ,IHW,P)
      STOP
      END
C
C
C
C
C     THE FOLLOWING SUBROUTINE GENERATES A LINEAR HORIZONTAL PLANE
C     WAVE AT THE TOP OF THE GRID.
C
      SUBROUTINE INPUT 1(NX,PS)
      COMPLEX PS(NX),CEXP
      DO 20 J=1,NX
   20 PS(J)=(1000.,0.)
      RETURN
      END
C
C
C
C
C     THE FOLLOWING SUBROUTINE GENERATES A CIRCULAR CONVERGENT PLANE
C     WAVE AT THE TOP OF THE GRID. THE CENTER OF CURVATURE IS LOCATED
C     AT THE CENTER OF THE GRID.
C
      SUBROUTINE INPUT 2(NX,NZ,DX,DZ,M,PS)
      REAL M
      COMPLEX PS(NX),CEXP,CI
      CI=(0.,1.)
      DO 20 J=1,NX
      R=SQRT((NZ*DZ/2.)**2+((J-NX/2.)*DX)**2)
   20 PS(J)=1000.*CEXP(CI*(M*R+3.1416/4.))/R
       ETURN
      END
C
C
C     THE FOLLOWING SUBROUTINE GENERATES A CIRCULAR DIVERGENT PLANE
C     WAVE WITH ITS CENTER OF CURVATURE BEING AT THE CENTER OF THE
C     GRID.
C
      SUBROUTINE INPUT 3(NX,NZ,DX,DZ,M,PS)
      REAL M
      COMPLEX PS(NX),CEXP,CI
      CI=(0.,-1.)
      DO 20 J=1,NX
      R=SQRT((NZ*DZ/2.)**2+((J-NX/2.)*DX)**2)
   20 PS(J)=1000.*CEXP(CI*(M*R+3.1416/4.))/R
```

```
      RETURN
      END
C
C
C     NEXT SUBROUTINE GENERATES AN INHOMOGENEITY OF TRIANGULAR
C      SHAPE WITH ITS BASE BEING HORIZONTAL AND ABOVE THE HYPOTENUSA.
C      J1,J2,I1 AND I2 ARE THE GRID COORDINATES OF THE TRIANGLE'S CORNERS.
C
      SUBROUTINE MEDIUM 1(E,M,NX,DX,DZ,I,A,X,Y,XS,XSS,AMP)
      INTEGER AMP
      REAL M
      COMPLEX A(NX),CMPLX
      J1=15*AMP
      J2=55*AMP
      I1=15*AMP
      I2=55*AMP
      IF(I.GE.I1.AND.I.LE.I2) GO TO 20
      DO 15 J=1,70
   15 A(J)=CMPLX(XSS,Y)
      GO TO 30
   20 DO 25 J=1,70
      IF(J.LE.J1.OR.J.GE.J2;OR.I.LE.J) GO TO 35
      A(J)=CMPLX(XS,Y)
      GO TO 25
   35 A(J)=CMPLX(XSS,Y)
   25 CONTINUE
   30 RETURN
      END
C     FOLLOWING SUBROUTINE GENERATES AN INHOMOGENEITY OF TRIANGULAR
C     SHAPE WITH ITS BASE BEING HORIZONTAL AND BELOW THE HYPOTENUSA.
C
      SUBROUTINE MEDIUM 2(E,M,NX,DX,DZ,I,A,X,Y,XS,XSS,AMP)
      INTEGER AMP
      REAL M
      COMPLEX A(NX),CMPLX
      J1=15*AMP
      J2=55*AMP
      I1=15*AMP
      I2=55*AMP
      IF(I.GE.I1.AND.I.LE.I2) GO TO 20
      DO 15 J=1,NX
   15 A(J)=CMPLX(XSS,Y)
      GO TO 30
   20 DO 25 J=1,NX
      IF(J.LE.J1.OR.J.GE.J2.OR.I.LE.J) GO TO 35
      A(J)=CMPLX(XS,Y)
      GO TO 25
   35 A(J)=CMPLX(XSS,Y)
   25 CONTINUE
   30 RETURN
      END
C
C
```

```
C
C     NEXT SUBROUTINE GENERATES AN INHOMOGENEOUS REGION WITH THE SHAPE
C     OF A HORIZONTAL PLANOCONVEX LENSE. J0 AND I0 ARE THE COORDINATES
C     OF THE LENSE'S CENTER OF CURVATURE. R IS ITS RADIUS AND IL IS THE
C     Z-COORDINATE OF THE PLANE FACE.
C
      SUBROUTINE MEDIUM 3(E,M,NX,DX,DZ,I,A,X,Y,XS,XSS,AMP)
      INTEGER AMP,R
      REAL M
      COMPLEX A(NX),CMPLX
      J0=35*AMP
      I0=55*AMP
      IL=50*AMP
      R=30*AMP
      T=R**2-(IL-I0)**2
      JL=SQRT(T)
      IF(I.GE.I0-R.AND.I.LE.IL) GO TO 20
      DO 15 J=1,NX
   15 A(J)=CMPLX(XSS,Y)
      GO TO 30
   20 DO 25 J=1,NX
      S=R**2-(J-J0)**2
      IF(J.LE.J0-JL.OR.J.GE:J0+JL.OR.I.LE.I0-SQRT(S)) GO TO 35
      A(J)=CMPLX(XS,Y)
      GO TO 25
   35 A(J)=CMPLX(XSS,Y)
   25 CONTINUE
   30 RETURN
      END
C
C
C
C
      SUBROUTINE CTRI(LX,CDIAG,CRIGHT,CANS,CSUB,CS,CBETA,CGAMMA)
      COMPLEX CSUB(LX),CDIAG(LX),CRIGHT(LX),CANS(LX),CBETA(LX)
      COMPLEX CGAMMA(LX),CDEN,CONJG,CS
      LXM1=LX-1
      CSUB(1)= (CRIGHT(1)*CONJG(CDIAG(1))+CRIGHT(2))*CS
      CSUB(LX)= (CRIGHT(LXM1)+CRIGHT(LX)*CONJG(CDIAG(LX)))*CS
      DO 30 J=2,LXM1
   30 CSUB(J)= (CRIGHT(J-1)+CRIGHT(J)*CONJG(CDIAG(J))+CRIGHT(J+1))*CS
      CBETA(1)=(1.,0.)
      CGAMMA(1)=(0.,0.)
      DO 10 I=2,LXM1
      CDEN=CDIAG(I)+CBETA(I-1)
      CBETA(I)=-(1.,0.)/CDEN
   10 CGAMMA(I)=(CSUB(I)-CGAMMA(I-1))/CDEN
      CANS(LX)=CGAMMA(LXM1)/(1.-CBETA(LXM1))
      DO 20 J=1,LXM1
      I=LX-J
   20 CANS(I)=CBETA(I)*CANS(I+1)+CGAMMA(I)
      RETURN
      END
C
```

```
C
C
C
      SUBROUTINE OUT(NX,NZ,P,DX,DZ,VEPP)
      DIMENSION P(NX,NZ)
      DIMENSION ICHAR(21),LINE(120)
      DATA ICHAR/'HHHH','GGGG','FFFF','EEEE','DDDD','CCCC','BBBB',
     1'AAAA','    ','    ','1111','2222','3333','4444','5555',
     2 '6666','7777','8888','9999','0000','****'/
      VR=(DX/DZ)*VEPP
      WRITE(6,66) VR
   66 FORMAT('-','VERTICAL EXAGERATION OF PRINTER PLOT IS',F5.2)
      B=0.
      DO 30 ID=1,NZ,3
      DO 30 IX=1,NX,3
      T=P(IX,ID)
   30 IF(ABS(T).GT.B) B=ABS(T)
      NXDONE=0
   40 NL=MIN0(120,NX-NXDONE)
      DO 20 ID=1,NZ
      DO 10 IL=1,NL
      IVAL=10.+P(IL+NXDONE,ID)*12./B
   10 LINE(IL)=ICHAR(MIN0(21,MAX0(1,IVAL)))
   20 WRITE(6,77) ID,(LINE(IL),IL=1,NL)
   77 FORMAT(I4,120A1)
      WRITE(6,71)
   71 FORMAT('1')
      NXDONE=NXDONE+120
      IF(NXDONE.LT.NX) GO TO 40
      RETURN
      END
C
C
C
C
      SUBROUTINE VOUT(VERTEX,VXOUT,NX,ND,IQ)
      REAL*4 IQ(NX,ND)
      INTEGER*4 IBITS(32)
      LOGICAL*4 BITS(32),LINE(18),MASKS(32),BASE,LIGHT(18)
      EQUIVALENCE (BITS(1),IBITS(1))
      DATA BITS(1),LIGHT/Z80000000,18*Z00000000/
      DO 15 I=1,20
   15 CALL VLINE('ESTEVEZ,BIN 603',15)
      DO 5 I=1,250
 5    CALL WRITER(LIGHT,70)
      DO 10 I=2,32
 10   IBITS(I)=2**(32-I)
      NSHIFT=(NX+17.499)/17.5
      NMAG=32/NSHIFT
      NDOT=AMAX1(1.,((NMAG+.5)/VERTEX)*VXOUT)
      VV=(VERTEX*NDOT)/NMAG
      WRITE(6,66) VV
 66   FORMAT(' VERTICAL EXAGERATION ON VERSATEC PLOT IS', F5.2)
      BASE=LIGHT(1)
```

```
      DO 20 I=1,32
      BASE=BASE.OR.BITS(I)
      MASKS(I)=BASE
 20   IF(MOD(I,NMAG).EQ.0) BASE=LIGHT(1)
      SCALE=NMAG/BIGEST(NX,ND,IQ)/NDOT/2.
      BIAS=0.5*NMAG+.499
      NDM1=ND-1
      DO 70 ID=1,NDM1
      DO 70 IDOT=1,NDOT
      DO 40 I=1,18
 40   LINE(I)=LIGHT(1)
      MSHIFT=0
      DO 60 ISHIFT=1,NSHIFT
      IL=1
      DO 50 IX=ISHIFT,NX,NSHIFT
      VAL=IDOT*IQ(IX,ID+1)+(NDOT-IDOT)*IQ(IX,ID)
      IMAG=BIAS+VAL*SCALE
      IF(IMAG.LE.0) GO TO 50
      IMAG=MIN0(NMAG,IMAG)
      LINE(IL)=LINE(IL).OR.MASKS(IMAG+MSHIFT)
 50   IL=IL+1
 60   MSHIFT=MSHIFT+NMAG
 70   CALL WRITER(LINE,70)
      DO 72 I=1,200
 72   CALL WRITER(LIGHT,70)
      DO 25 I=1,20
   25 CALL VLINE('ESTEVEZ,BIN 603',15)
      RETURN
      END
C
C
C
C
      SUBROUTINE TOUT(CLIP,NXOUT,NZ, IHW,P)
      REAL P(NX,NZ)
      INTEGER*2 IHW(NZ)
      SCALE=(1024./BIGEST(NXOUT,NZ,P))*CLIP
      DO 260 I=1,NXOUT
      DO 270 J=1,NZ
  270 IHW(J)=P(I,J)*SCALE
  260 WRITE(8,800) (IHW(J),J=1,NZ)
  800 FORMAT(4(255A2))
      END FILE 8
      REWIND 8
      RETURN
      END
C
C
C
C
      FUNCTION BIGEST(NX,ND,IQ)
      REAL*4 IQ(NX,ND)
      B=0.
      DO 30 ID=1,ND,3
```

```
      DO 30 IX=1,NX,3
      T=IQ(IX,ID)
   30 IF(ABS(T).GT.B) B=ABS(T)
      BIGEST=B
      RETURN
      END
```

References

Claerbout, J. F., 1970a, Coarse grid calculations of waves in inhomogeneous media with application to delineation of complicated seismic structures: Geophysics, v. 35, p. 407-418.

Claerbout, J. F., 1970b, Numerical holography in acoustical holography, v. 13, edited by A. F. Metherell, N. Y. Plenum Press.

Claerbout, J. F., 1973, Class notes at Stanford University, Geophysics Department.